8 KINEMATIC CHAINS WITH CONTINUOUS FLEXIBLE LINKS

8.1 Transverse vibrations of a flexible link

Often a kinematic chain consists in part of continuous elastic components supported by rigid bodies. Small motions of the components relative to the rigid bodies generally are governed by partial differential equations. In such cases these equations cannot be solved by the method of separation of variables.

Figure 8.1 shows a cantilever beam B of length L, constant flexural rigidity EI and constant mass per unit length ρ. When B is supported by a rigid body fixed in a frame, small flexural vibrations of B are governed by the equation

$$EI\frac{\partial^4 y(x,t)}{\partial x^4} + \rho\frac{\partial^2 y(x,t)}{\partial t^2} = 0, \tag{8.1}$$

and by the boundary conditions

$$y(0,t) = y^{'}(0,t) = y^{''}(L,t) = y^{'''}(L,t) = 0. \tag{8.2}$$

The general solution of Eq. (8.1) that satisfies Eq. (8.2) can be expressed as

$$y(x,t) = \sum_{i=1}^{\infty} \Phi_i(x)\, q_i(t)\,, \tag{8.3}$$

where $\Phi_i(x)$ and $q_i(t)$ are functions of x and t, respectively, defined as

$$\Phi_i = \cosh\frac{\lambda_i x}{L} - \cos\frac{\lambda_i x}{L} - \frac{\cosh\lambda_i + \cos\lambda_i}{\sinh\lambda_i + \sin\lambda_i}\left(\sinh\frac{\lambda_i x}{L} - \sin\frac{\lambda_i x}{L}\right), \tag{8.4}$$

and

$$q_i = \alpha_i\cos p_i t + \beta_i \sin p_i t, \tag{8.5}$$

where λ_i, $i = 1, \ldots, \infty$ are the consecutive roots of the transcendental equation

$$\cos\lambda\cosh\lambda + 1 = 0, \tag{8.6}$$

while

$$p_i = \left(\frac{\lambda_i}{L}\right)^2 \left(\frac{EI}{\rho}\right)^{1/2}, \tag{8.7}$$



Figure 8.1

and α_i and β_i are constants that depends upon initial conditions.
The functions $\Phi_i(x)$ satisfy the orthogonality relations

$$\int_0^L \Phi_i \, \Phi_j \, \rho \, dx = m \, \delta_{ij} \qquad (i, j = 1, \ldots, \infty), \tag{8.8}$$

and

$$EI \int_0^L \Phi_i'' \, \Phi_j'' \, d\, x = p_i^2 \, m \, \delta_{ij} \qquad (i, j = 1, \ldots, \infty), \tag{8.9}$$

where m is the mass of the beam and δ_{ij} is the Kronecker delta.

8.2 Equations of motion for a flexible link

In Fig. 8.2, a schematic representation of a kinematic chain is given. The system is formed by a rigid body RB that supports a uniform cantilever beam B of length L, flexural rigidity EI, and mass per unit length ρ. Only planar motions of the kinematic chain in a fixed reference frame (0) of unit vectors $[\mathbf{\imath}_0, \, \mathbf{\jmath}_0, \, \mathbf{k}_0]$ will be considered.
To characterize the instantaneous configuration of the rigid body RB, generalized coordinates q_1, q_2, q_3 are employed. The first generalized coordinate q_1 denotes the distance from C_R, the mass center of RB, to the horizontal axis of the reference frame (0). The generalized coordinate q_2 denotes the distance from C_R, to the vertical axis of (0). The last generalized coordinate q_3, ($s_3 = \sin q_3$, $c_3 = \cos q_3$), designates the radian measure of the rotation angle between RB and the horizontal axis.
Generalized speeds u_1, u_2, and u_3, used to characterize the motion of RB in (0), are defined as

$$u_1 = \mathbf{v}_{C_R} \cdot \mathbf{\imath}, \;\; u_2 = \mathbf{v}_{C_R} \cdot \mathbf{\jmath}, \;\; u_3 = \boldsymbol{\omega}_{R0} \cdot \mathbf{k}, \tag{8.10}$$

where $\mathbf{v}_{C_R}$ is the velocity in (0) of the mass center C_R of RB, $\boldsymbol{\omega}_{R0}$ is the angular velocity of RB in (0), and $[\mathbf{\imath}, \, \mathbf{\jmath}, \, \mathbf{k}]$ form a dextral set of mutually perpendicular unit vectors fixed in RB and directed as shown in Fig. 8.2. It follows immediately that

$$\mathbf{v}_{C_R} = u_1 \, \mathbf{\imath} + u_2 \, \mathbf{\jmath}, \qquad \boldsymbol{\omega}_{R0} = u_3 \mathbf{k}. \tag{8.11}$$

The unit vectors $\mathbf{\imath}_0, \mathbf{\jmath}_0, \mathbf{k}_0$ can be expressed as

$$\begin{aligned} \mathbf{\imath}_0 &= c_3 \, \mathbf{\imath} - s_3 \, \mathbf{\jmath}, \\ \mathbf{\jmath}_0 &= s_3 \, \mathbf{\imath} + c_3 \, \mathbf{\jmath}, \\ \mathbf{k}_0 &= \mathbf{k}. \end{aligned} \tag{8.12}$$



Figure 8.2

The velocity of C_R in (0) is

$$\begin{aligned} \mathbf{v}_{C_R} &= \dot{q}_1\, \mathbf{\imath}_0 + \dot{q}_2\, \mathbf{\jmath}_0 \\ &= (\dot{q}_1\, c_3 + \dot{q}_2\, s_3)\mathbf{\imath} + (-\dot{q}_1\, s_3 + \dot{q}_2\, c_3)\mathbf{\jmath}. \end{aligned} \tag{8.13}$$

From Eqs. (8.11) and (8.13) follows that

$$\begin{aligned} u_1 &= \dot{q}_1\, c_3 + \dot{q}_2\, s_3, \\ u_2 &= -\dot{q}_1\, s_3 + \dot{q}_2\, c_3, \\ u_3 &= \dot{q}_3. \end{aligned} \tag{8.14}$$

Equation (8.14) can be solved uniquely for $\dot{q}_1, \dot{q}_2, \dot{q}_3$, and thus u_1, u_2, u_3 form a set of generalized speeds for the RB.

Kinematics

Deformations of the cantilever beam B can be discussed in terms of the displacement $y(x,t)$ of a generic point P on the beam B. The point P is situated at a distance x from the point Q, the point at which B is attached to RB. The displacement y can be expressed as

$$y(x,t) = \sum_{i=1}^{n} \Phi_i(x)\, q_{3+i}(t), \tag{8.15}$$

where $\Phi_i(x)$ is a totally unrestricted function of x, $q_{3+i}(t)$ is an equally unrestricted function of t, and n is any positive integer. Generalized speeds u_{3+i}, $i = 1, \ldots, n$ are introduced as

$$u_{3+i} = \dot{q}_{3+i}, \qquad i = 1, \ldots, n. \tag{8.16}$$

The velocity of Q in (0) is

$$\begin{aligned} \mathbf{v}_Q &= \mathbf{v}_{C_R} + \boldsymbol{\omega}_{R0} \times b\, \mathbf{\imath} \\ &= u_1\, \mathbf{\imath} + u_2\, \mathbf{\jmath} + \begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & u_3 \\ b & 0 & 0 \end{vmatrix} = u_1 \mathbf{\imath} + (u_2 + b\, u_3)\mathbf{\jmath}, \end{aligned} \tag{8.17}$$

where b is the distance from C_R to Q.
The velocity of any point P of the elastic link B in (0) is

$$\mathbf{v}_P = \mathbf{v}_{C_R} + \frac{\partial}{\partial t}\left[(b+x)\, \mathbf{\imath} + y\, \mathbf{\jmath}\right] + \boldsymbol{\omega}_{R0} \times \left[(b+x)\, \mathbf{\imath} + y\, \mathbf{\jmath}\right]$$

$$
\begin{aligned}
&= u_1\,\mathbf{\imath} + u_2\,\mathbf{\jmath} + \dot{y}\,\mathbf{\jmath} + \begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & u_3 \\ b+x & y & 0 \end{vmatrix} \\
&= (u_1 - u_3\,y)\mathbf{\imath} + [u_2 + (b+x)\,u_3 + \dot{y}]\mathbf{\jmath} \\
&= \left(u_1 - u_3 \sum_{i=1}^{n} \Phi_i\, q_{3+i}\right)\mathbf{\imath} \\
&+ \left[u_2 + (b+x)\,u_3 + \sum_{i=1}^{n} \Phi_i\, u_{3+i}\right]\mathbf{\jmath}. \qquad (8.18)
\end{aligned}
$$

The velocity of the midpoint C_B of the uniform elastic link B in (0) is

$$
\begin{aligned}
\mathbf{v}_{C_B} &= \mathbf{v}_P(x = \frac{L}{2},\, t) = \left(u_1 - u_3 \sum_{i=1}^{n} \Phi_i\, q_{3+i}\right)\mathbf{\imath} \\
&+ \left[u_2 + (b + 0.5\,L)\,u_3 + \sum_{i=1}^{n} \Phi_i\, u_{3+i}\right]\mathbf{\jmath}. \qquad (8.19)
\end{aligned}
$$

The angular acceleration of RB in the reference frame (0) is

$$
\boldsymbol{\alpha}_{R0} = \dot{\boldsymbol{\omega}}_{R0} = \dot{u}_3\mathbf{k}. \qquad (8.20)
$$

The linear acceleration of C_R in the reference frame (0) is

$$
\begin{aligned}
\mathbf{a}_{C_R} &= \frac{\partial}{\partial t}\mathbf{v}_{C_R} + \boldsymbol{\omega}_{R0} \times \mathbf{v}_{C_R} \\
&= \dot{u}_1\,\mathbf{\imath} + \dot{u}_2\,\mathbf{\jmath} + \begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & u_3 \\ u_1 & u_2 & 0 \end{vmatrix} \\
&= (\dot{u}_1 - u_2\,u_3)\mathbf{\imath} + (\dot{u}_2 + u_3\,u_1)\mathbf{\jmath}.
\end{aligned}
$$

The acceleration of point P in the reference frame (0) is

$$
\begin{aligned}
\mathbf{a}_P &= \frac{\partial}{\partial t}\mathbf{v}_P + \boldsymbol{\omega}_{R0} \times \mathbf{v}_P \\
&= \left[\dot{u}_1 - u_2u_3 - (b+x)u_3^2 - \sum_{i=1}^{n} \Phi_i(\dot{u}_3q_{3+i} + 2u_3u_{3+i})\right]\mathbf{\imath} \\
&+ \left[\dot{u}_2 + u_3u_1 + (b+x)\dot{u}_3 + \sum_{i=1}^{n} \Phi_i(\dot{u}_{3+i} - u_3^2q_{3+i})\right]\mathbf{\jmath}. \qquad (8.21)
\end{aligned}
$$

Generalized inertia forces

If m_R and I_z are the mass of RB and the moment of inertia of RB about a line passing through C_R and parallel to $\mathbf{k}$, then the generalized inertia force F_r^* is given by

$$\begin{aligned} F_r^* \quad = \quad & \frac{\partial \boldsymbol{\omega}_{R0}}{\partial u_r} \cdot (-I_z\, \boldsymbol{\alpha}_{R0}) + \frac{\partial \mathbf{v}_{C_R}}{\partial u_r} \cdot (-m_R\, \mathbf{a}_{C_R}) + \\ & \int_o^L \frac{\partial \mathbf{v}_P}{\partial u_r} \cdot (-\mathbf{a}_P)\, \rho\, d\, x, \quad r = 1, \ldots, 3+n. \end{aligned} \tag{8.22}$$

The constants m_B, e_B, I_B, E_i, F_i, and G_{ij} are defined as

$$m_B{=}\int_0^L \rho dx, \;\; e_B{=}\int_0^L x\rho dx, \;\; I_B{=}\int_0^L x^2\rho dx,$$

$$E_i{=}\int_0^L \Phi_i\, \rho\; dx, \;\; F_i{=}\int_0^L x\Phi_i\, \rho\; dx, \;\; G_{ij}{=}\int_0^L \Phi_i\, \Phi_j \rho\; dx,$$
$$i, j = 1, \ldots, n.$$

Equation (8.22) then leads to

$$\begin{aligned}
& F_1^* = -\,(m_R + m_B)(\dot{u}_1 - u_2 u_3) + \dot{u}_3 \sum_{i=1}^n E_i q_{3+i} \\
& +2u_3 \sum_{i=1}^n E_i u_{3+i} + u_3^2 (b m_B + e_B), \\
& F_2^* = -\,(m_R + m_B)(\dot{u}_2 + u_3 u_1) - \sum_{i=1}^n E_i \dot{u}_{3+i} \\
& -\dot{u}_3 (b m_B + e_B) + u_3^2 \sum_{i=1}^n E_i q_{3+i}, \\
& F_3^* = (\dot{u}_1 - u_2 u_3) \sum_{i=1}^n E_i q_{3+i} - (\dot{u}_2 + u_3 u_1)(b m_B + e_B) \\
& -\dot{u}_3 (b^2 m_B + 2b e_B + I_B + I_z) - \sum_{i=1}^n \dot{u}_{3+i} (b E_i + F_i) \\
& -2u_3 \sum_{i=1}^n \sum_{k=1}^n G_{ik} q_{3+i} u_{3+k} - \dot{u}_3 \sum_{i=1}^n \sum_{k=1}^n G_{ik} q_{3+i} q_{3+k}, \\
& F_{3+j}^* = -\,\dot{u}_2 E_j - \sum_{i=1}^n G_{ij} \dot{u}_{3+i} - \dot{u}_3 (b E_j + F_j) - u_3 u_1 E_j \\
& +u_3^2 \sum_{i=1}^n G_{ij} q_{3+i}, \qquad j = 1, \ldots, n.
\end{aligned} \tag{8.23}$$

Generalized active forces

The contributions to the generalized active forces are made by the internal forces, and by the gravitational forces exerted. The internal forces are considered first. The force $d\mathbf{F}$ is the force exerted on a generic differential element of B

$$d\mathbf{F} = -\frac{\partial V(x,t)}{\partial x}\, dx\ \mathbf{\jmath}, \tag{8.24}$$

where $V(x,t)$ is the shear at point P. If rotatory inertia is neglected, then $V(x,t)$ may be expressed in terms of the bending moment $M(x,t)$ as

$$V(x,t) = \frac{\partial M(x,t)}{\partial x}. \tag{8.25}$$

Since

$$M = EI\frac{\partial^2 y}{\partial x^2}\ , \tag{8.26}$$

Eqs. (8.24),(8.26) yield

$$d\mathbf{F} = -\frac{\partial^2}{\partial x^2}\left(EI\frac{\partial^2 y}{\partial x^2}\right)\, dx\ \mathbf{\jmath}. \tag{8.27}$$

The system of forces exerted on the rigid body RB by the elastic beam B is equivalent to a couple of torque $M(0,t)\mathbf{k}$ together with a force - $V(0,t)\mathbf{\jmath}$ applied at point Q. Hence, $(F_r)_I$, the contribution of the internal forces to the generalized active force F_r, is given by

$$\begin{aligned}(F_r)_I &= \frac{\partial \boldsymbol{\omega}_{R0}}{\partial u_r}\cdot M(0,t)\mathbf{k} - \frac{\partial \mathbf{v}_Q}{\partial u_r}\cdot V(0,t)\mathbf{\jmath} - \\ &\quad \int_0^L \frac{\partial \mathbf{v}_P}{\partial u_r}\cdot \mathbf{\jmath}\frac{\partial^2}{\partial x^2}\left(EI\frac{\partial^2 y}{\partial x^2}\right)\, dx \\ &= EI\left(\frac{\partial \boldsymbol{\omega}_{R0}}{\partial u_r}\cdot \mathbf{k}\frac{\partial^2 y(0,t)}{\partial x^2} - \frac{\partial \mathbf{v}_Q}{\partial u_r}\cdot \mathbf{\jmath}\frac{\partial^3 y(0,t)}{\partial x^3}\right) - \\ &\quad \int_0^L \frac{\partial \mathbf{v}_P}{\partial u_r}\cdot \mathbf{\jmath}\frac{\partial^2}{\partial x^2}\left(EI\frac{\partial^2 y(x,t)}{\partial x^2}\right)\, dx,\ \ r = 1,\ldots,3+n.\end{aligned} \tag{8.28}$$

which leads to

$$(F_1)_I = 0,$$

$$
\begin{aligned}
(F_2)_I &= -\sum_{i=1}^{n} q_{3+i} \left[(EI\Phi_i^{'''})_{x=0} + \int_0^L (EI\Phi_i^{''})^{''} \, dx \right], \\
(F_3)_I &= b\,(F_2)_I + \sum_{i=1}^{n} q_{3+i} \left[(EI\Phi_i^{''})_{x=0} - \int_0^L x(EI\Phi_i^{''})^{''} \, dx \right], \\
(F_{3+j})_I &= -\sum_{i=1}^{n} q_{3+i} \int_0^L \Phi_j (EI\Phi_i^{''})^{''} \, dx, \qquad j = 1, \ldots, n. \qquad (8.29)
\end{aligned}
$$

The restrictions on Φ_i to ensure that y and $y^{'}$ vanish at $x = 0$ while M and V vanish at $x = L$ are

$$
\Phi_i(0) = \Phi_i^{'}(0) = \Phi_i^{''}(L) = \Phi_i^{'''}(L) = 0, \qquad i = 1, \ldots, n. \qquad (8.30)
$$

When the integrations are carried out, the following expressions result for the contribution of the internal forces to the generalized active forces

$$
\begin{aligned}
&(F_1)_I = (F_2)_I = (F_3)_I = 0, \\
&(F_{3+j})_I = -\sum_{i=1}^{n} H_{ij} q_{3+i}, \qquad j = 1, \ldots n, \qquad (8.31)
\end{aligned}
$$

where H_{ij} is defined as

$$
H_{ij} = \int_0^L EI\Phi_i^{''}\Phi_j^{''} \, dx, \qquad i, j = 1, \ldots, n. \qquad (8.32)
$$

The gravitational forces exerted on RB and B by the Earth, are denoted by $\mathbf{G}_R$, and $\mathbf{G}_B$, and can be expressed as

$$
\begin{aligned}
\mathbf{G}_R &= -m_R\, g \mathbf{J}_0 = -m_R\, g\, (s_3\, \mathbf{\imath} + c_3\, \mathbf{\jmath}), \\
\mathbf{G}_B &= -m_B\, g\, \mathbf{J}_0 = -m_B\, g\, (s_3\, \mathbf{\imath} + c_3\, \mathbf{\jmath}). \qquad (8.33)
\end{aligned}
$$

The contribution to F_r of the gravitational forces is

$$
(F_r)_G = \frac{\partial \mathbf{v}_{C_R}}{\partial u_r} \cdot \mathbf{G}_R + \frac{\partial \mathbf{v}_{C_B}}{\partial u_r} \cdot \mathbf{G}_B, \;\; r = 1, \ldots, 3+n. \qquad (8.34)
$$

The generalized active forces are

$$
F_r = (F_r)_I + (F_r)_G, \;\; r = 1, \ldots, 3+n. \qquad (8.35)
$$

To arrive at the dynamical equations governing the system, all that remains to be done is to substitute into Kane's dynamical equations, namely,

$$
F_r^* + F_r = 0, \;\; r = 1, \ldots, 3+n. \qquad (8.36)
$$

In[13]:=
```
Off[General::spell]
Off[General::spell1]

(* FLEXIBLE BEAM CANTILEVED IN A RIGID BODY FALLING
   UNDER GRAVITY *)

Apply [ Clear, Names ["Global`*" ]];

(* (0) - fixed reference frame attached to the
ground with the unit vectors {i0, j0, k0} *)
(* (1) - mobile reference frame attached to the
rigid body base RB with the unit vectors {i, j, k}*)

(*Cross[xx_ , yy_] :={xx[[2]] yy[[3]]-xx[[3]] yy[[2]],
                 xx[[3]] yy[[1]]-xx[[1]] yy[[3]],
                 xx[[1]] yy[[2]]-xx[[2]] yy[[1]]}; *)

(* initial data *)
L = 1.;
b = 0.1;
c = 2 b;
d = 0.3;
mR = 1.;
lam1 = 1.875;
ro = 0.21991;
mB = L ro;
EI = 35.3;
Iz = 1 / 12. mR (c^2 + d^2);
g = 9.81;

(* transformation matrix from (0) to (1) *)
R01 = {{Cos[q3[t]],  -Sin[q3[t]],    0},
       { Sin[q3[t]],  Cos[q3[t]],    0},
       {  0,         0,    1}};

(* kinematic eqs. in generalized speeds  *)
rule = {q1'[t] -> u1[t] Cos[q3[t]] - u2[t] Sin[q3[t]],
       q2'[t] -> u1[t] Sin[q3[t]] + u2[t] Cos[q3[t]],
       q3'[t] -> u3[t],
       q4'[t] -> u4[t]};

(* angular velocity of RB in (0)*)
wR = {0, 0, u3[t]};

(* linear velocity of mass center CR of RB in (0)
expressed in terms of (1) {i,j,k} *)
vCR = {u1[t], u2[t], 0};

(* linear velocity vQ of point Q on RB in (0)
expressed in terms of (1) {i,j,k} *)
rQ = {b, 0, 0};
vQ = vCR + Cross[wR, rQ];
Fi1 = (Cosh[lam1 x / L] - Cos[lam1 x / L] -
(Cosh[lam1] + Cos[lam1]) / (Sinh[lam1] + Sin[lam1])
(Sinh[lam1 x / L] - Sin[lam1 x / L]));

(* elastic displacement of an arbitrary point P on the
```

```
elastic link B in (1) expressed in terms of (1) {i,j,k}*)
yP1 = q4[t] Fi1;

(* linear velocity of an arbitrary point P on the elastic
link in (0) expressed in terms of (1) {i,j,k} *)
rP = {b + x, yP1, 0};
vP = vCR + D[rP, t] + Cross[ wR, rP] /. rule;
FiCB = Fi1 /. x -> L / 2.;

(* elastic displacement of the midpiont CB on the
elastic link B in (1) expressed in terms of (1) {i,j,k}*)
yCB = q4[t] FiCB;

(* linear velocity of a midpiont CB on the elastic
link B in (0) expressed in terms of (1) {i,j,k} *)
rCB = {b + L / 2, yCB , 0};
vCB = vCR + D[rCB, t] + Cross[wR, rCB] /. rule;

(* angular acceleration of RB in (0) expressed in terms
of (1) {i,j,k} *)
alphaR = D[wR, t];

(* linear acceleration of CR of RB in (0) expressed
in terms of (1) {i,j,k} *)
aCR = Simplify [ D[vCR, t] + Cross[wR, vCR] /. rule];

(* linear acceleration of an arbitrary point P on link
B in (0) expressed in terms of (1) {i,j,k} *)
aP = D[vP, t] + Cross[wR, vP ] /. rule;

(* generalized inertia forces  *)

inte1 = ExpandAll[D[vP, u1[t]] . (-aP) ro];
inte2 = ExpandAll[D[vP, u2[t]] . (-aP) ro];
inte3 = ExpandAll[D[vP, u3[t]] . (-aP) ro];
inte4 = ExpandAll[D[vP, u4[t]] . (-aP) ro];

Integrala1 = Chop[Integrate[inte1, {x, 0, L}]];
Integrala2 = Chop[Integrate[inte2, {x, 0, L}]];
Integrala3 = Chop[Integrate[inte3, {x, 0, L}]];
Integrala4 = Chop[Integrate[inte4, {x, 0, L}]];

Fin1 = D[wR, u1[t]] . (-Iz alphaR) +
D[vCR, u1[t]] . (-mR aCR) + Integrala1;

Fin2 = D[wR, u2[t]] . (-Iz alphaR) +
D[vCR, u2[t]] . (-mR aCR) + Integrala2;

Fin3 = D[wR, u3[t]] . (-Iz alphaR) +
D[vCR, u3[t]] . (-mR aCR) + Integrala3;

Fin4 = D[wR, u4[t]] . (-Iz alphaR) +
D[vCR, u4[t]] . (-mR aCR) + Integrala4;

(* generalized active forces  *)

(* gravitational forces  *)
```

```
GR = {0, -mR g , 0} .R01;
GB = {0, -mB g , 0} .R01;
F1G = D[vCR, u1[t]] .GR + D[vCB, u1[t]] .GB;
F2G = D[vCR, u2[t]] .GR + D[vCB, u2[t]] .GB;
F3G = D[vCR, u3[t]] .GR + D[vCB, u3[t]] .GB;
F4G = D[vCR, u4[t]] .GR + D[vCB, u4[t]] .GB;

M = EI D[D[yP1, x], x];
M0 = {0 , 0 , M /. x -> 0};
V = D[M, x];
V0 = {0 , V /. x -> 0 , 0};
F = {0 , D[D[M, x], x] , 0};
integ1 = ExpandAll[D[vP, u1[t] ] . F];
integ2 = ExpandAll[D[vP, u2[t]] . F];
integ3 = ExpandAll[D[vP, u3[t]] . F];
integ4 = ExpandAll[D[vP, u4[t]] . F];

Integral1 = Chop[Integrate[integ1, {x, 0, L}]];
Integral2 = Chop[Integrate[integ2, {x, 0, L}]];
Integral3 = Chop[Integrate[integ3, {x, 0, L}]];
Integral4 = Chop[Integrate[integ4, {x, 0, L}]];

F1I = D[wR, u1[t]] .M0 - D[vQ, u1[t]] .V0 - Integral1;

F2I = D[wR, u2[t]] .M0 - D[vQ, u2[t]] .V0 - Integral2;

F3I = D[wR, u3[t]] .M0 - D[vQ, u3[t]] .V0 - Integral3;

F4I = D[wR, u4[t]] .M0 - D[vQ, u4[t]] .V0 - Integral4;

e1 = Fin1 + F1I + F1G;
e2 = Fin2 + F2I + F2G;
e3 = Fin3 + F3I + F3G;
e4 = Fin4 + F4I + F4G;
e5 = -u3[t] + q3'[t];
e6 = -u4[t] + q4'[t];
(* numerical simulation *)

kane = NDSolve[
{e1 == 0, e2 == 0, e3 == 0, e4 == 0, e5 == 0, e6 == 0,
u1[0] == 0.0, u2[0] == 0.0, u3[0] == 0.0,
u4[0] == 0.0, q3[0] == 1.0, q4[0] == 0.0001},
{u1[t], u2[t], u3[t], u4[t], q3[t], q4[t]}, {t, 0.0, 1.0}];
Plot [Evaluate[u1[t] /. kane], {t, 0.0, 1.0},
PlotRange -> {All, All},
AxesLabel -> { "t[s]", "u1[m/s]" }];
Plot [Evaluate[u2[t] /. kane], {t, 0.0, 1.0},
PlotRange -> {All, All},
AxesLabel -> { "t[s]", "u2[m/s]" }];
Plot[Evaluate[u3[t] /. kane], {t, 0.0, 1.0},
PlotRange -> {All, All },
AxesLabel -> {"t[s]", "u3[rad/s]"}];
Plot[Evaluate[u4[t] /. kane], {t, 0.0, 1.0},
PlotRange -> {All, All },
AxesLabel -> {"t[s]", "u4[m/s]"}];
Plot[Evaluate[q3[t] /. kane], {t, 0.0, 1.0},
```

```
PlotRange -> {All, All },
AxesLabel -> {"t[s]", "q3[m]"}];
Plot[Evaluate[q4[t] /. kane], {t, 0.0, 1.0},
PlotRange -> {All, All },
AxesLabel -> {"t[s]", "q4[m]"}];
```








u4[m/s]
0.02
0.01
-0.01
-0.02
0.2
0.4
0.6
0.8
1
t[s]


q3[m]
0.9999
0.9998
0.9997
0.9996
0.9995
0.9994
0.2
0.4
0.6
0.8
1
t[s]


q4[m]
0.00045
0.0004
0.00035
0.0003
0.00025
0.0002
0.00015
0.2
0.4
0.6
0.8
1
t[s]